6.b System issues (a concrete implementation)

Task model summary – 1

- Static task set
  - In Ada: all tasks declared at library level
- Tasks issue jobs repeatedly
  - Task cycle: activation, execution, suspension
    - Single activation point, no blocking (!)
- Real-time attributes
  - Activation
    - Periodic or cyclic: every T time units
    - Sporadic: at least T time units between consecutive events
  - Execution
    - Worst case execution time (WCET) assumed to be known
    - Deadline: D time units after activation

Task model summary – 2

- Task communication
  - Shared variables with mutually exclusive access
    - In Ada: protected objects with procedures and functions
  - No avoidance synchronization
    - Except for delivering release events to sporadic tasks
    - In Ada: PO with a single entry
- Scheduling model
  - Fixed-priority preemptive
    - In Ada: `FIFO_within_priorities`
- Access protocol for shared objects
  - Immediate ceiling priority
    - In Ada: `Ceiling_Locking`

The Ravenscar profile

- An Ada language profile is enforced by means of a configuration pragma
  - `pragma Profile (Ravenscar)`
- Equivalent to a set of language restrictions and three additional configuration pragmas
  - `pragma Dispatching_Policy (FIFO_Within_Priorities)`
  - `pragma Locking_Policy (Ceiling_Locking)`
  - `pragma Detect_Blocking`

Ravenscar restrictions

```
No_Abort_Statements,
No_Dynamic_Attachment,
No_Dynamic_Priorities,
No_Implicit_Heap_Allocations,
No_Local_Protected_Objects,
No_Local_Timing_Events,
No_Protected_Type_Allocators,
No_Relative_Delay,
No_Requeue_Statements,
No_Select_Statements,
No_Specific_Termination_Handlers,
No_Task_Allocators,
No_Task_Hierarchy,
No_Task_Termination,
Simple_Barriers,
Max_Entry_Queue_Length => 1,
Max_Protected_Entries => 1,
Max_Task_Entries => 0,
No_Dependence => Ada.Asynchronous_Task_Control,
No_Dependence => Ada.Calendar,
No_Dependence => Ada.Execution_Time.Group_Budget,
No_Dependence => Ada.Execution_Time.Timers,
No_Dependence => Ada.Task_Attributes
```


Restriction checking

- Luckily, almost all of the restrictions can be statically checked by the compiler
- A few restrictions can only be checked at run time
  - Potentially blocking operations in the bodies of protected operation
  - Priority ceiling violation
  - More than one call queued on a protected entry or a suspension object
  - Task termination

Potentially blocking operations

- Potentially "suspending" operations
  - Delay until statement
    - A true suspension
  - Protected entry call statement
    - Avoidance synchronization treated with the eggshell model
  - Transitive closure across procedure calls
    - Call on a subprogram whose body contains a potentially suspending operation
- `pragma Detect_Blocking` requires detection of potentially blocking operations
  - Exception `Program_Error` must be raised if detected at run time
  - Blocking need not be detected if it occurs in the domain of a foreign language (e.g. C)

Other run-time checks

- Priority ceiling violation
- More than one call waiting on a protected entry or a suspension object
  - Exception `Program_Error` must be raised in both cases
- Task termination
  - Program behavior must be documented
  - Possible effects include
    - Silent termination
    - Holding the task in a pre-terminated state
    - Execution on an application-defined termination handler
      - Use of the Ada.Task_Termination package (C.7.3)

Other restrictions

- Some restrictions on the sequential part of the language may be useful in conjunction with the Ravenscar profile
  - `No_Dispatch`
  - `No_IO`
  - `No_Recursion`
  - `No_Unchecked_Access`
  - `No_Allocators`
  - `No_Local_Allocators`
- See **ISO/IEC TR 15942**, *Guide for the use of the Ada Programming Language in High Integrity Systems* for details

Execution time measurement – 1

- The CPU time consumed by tasks during execution can be measured
  - And actually should if the WCET values used for feasibility analysis have to hold true (!)
- Per-task clocks can be defined
  - Set at 0 before task activation
  - The clock value increases as the task executes

Execution time measurement – 2

```
with Ada.Task_Identification;
with Ada.Real_Time; use Ada.Real_Time;
package Ada.Execution_Time is
   type CPU_Time is private;
   CPU_Time_First : constant CPU_Time;
   CPU_Time_Last  : constant CPU_Time;
   CPU_Time_Unit  : constant :=
        implementation-defined-real-number;
   CPU_Tick       : constant Time_Span;
   function Clock
     (T : Ada.Task_Identification.Task_Id
          := Ada.Task_Identification.Current_Task)
     return CPU_Time;
   ...
end Ada.Execution_Time;
```


Execution time timers – 1

- A user-defined event can be fired when a CPU clock reaches a specified value
  - An event handler is automatically invoked by the runtime
  - The handler is an (access to) a protected procedure
- Basic mechanism for execution-time monitoring

Execution time timers – 2

```
with System;
package Ada.Execution_Time.Timers is
   type Timer (T : not null access constant
                   Ada.Task_Identification.Task_Id) is
      tagged limited private;
   type Timer_Handler is
      access protected procedure (TM : in out Timer);
   Min_Handler_Ceiling : constant System.Any_Priority
      := implementation-defined;
   procedure Set_Handler (TM      : in out Timer;
                          In_Time : in Time_Span;
                          Handler : in Timer_Handler);
   procedure Set_Handler (TM      : in out Timer;
                          At_Time : in CPU_Time;
                          Handler : in Timer_Handler);
   ...
end Ada.Execution_Time.Timers;
```


Group budget – 1

- Groups of tasks with a global execution-time budget can be defined
  - Can be used to provide temporal isolation among groups of tasks
  - Basic mechanism for server-based scheduling

Group budget – 2

```
with System;
package Ada.Execution_Time.Group_Budgets is
  type Group_Budget is tagged limited private;
  type Group_Budget_Handler is
    access protected procedure (GB : in out Group_Budget);
  ...
  Min_Handler_Ceiling : constant System.Any_Priority :=
     implementation-defined;
  procedure Add_Task (GB : in out Group_Budget;
                      T  : in Ada.Task_Identification.Task_Id);
  ...
  procedure Replenish (GB : in out Group_Budget;
                       To : in Time_Span);
  procedure Add (GB       : in out Group_Budget;
                 Interval : in Time_Span);
  ...
  procedure Set_Handler (GB      : in out Group_Budget;
                         Handler : in Group_Budget_Handler);
  ...
end Ada.Execution_Time.Group_Budgets;
```


Timing events – 1

- Lightweight mechanism for defining code to be executed at a specified time
  - Does not require an application-level task
  - Analogous to interrupt handling
- The code is defined as an event handler
  - An (access to) a protected procedure
  - Directly invoked by the runtime system

Timing events – 2

```
package Ada.Real_Time.Timing_Events is
  type Timing_Event is tagged limited private;
  type Timing_Event_Handler is
    access protected procedure (Event : in out Timing_Event);
  procedure Set_Handler (Event   : in out Timing_Event;
                         At_Time : in Time;
                         Handler : in Timing_Event_Handler);
  ...
  procedure Cancel_Handler (Event     : in out Timing_Event;
                            Cancelled : out Boolean);
  ...
end Ada.Real_Time.Timing_Events;
```


Scheduling and dispatching policies

- Additional dispatching policies
  - Non preemptive
    - Run-to-completion semantics (per partition)
    - Built-in support provided
  - Round robin
    - Within specified priority band
    - Built-in support provided
    - Dispatch on quantum expiry is deferred until end of protected action
  - Earliest deadline first
    - Within specified priority band
    - Built-in support provided for relative and absolute "deadline"
    - EDF ordered ready queues
    - Guaranteed form of resource locking (preemption level + deadline)

Priority-band dispatching

- Mixed policies can coexist within a single partition
  - Priority specific dispatching policy can be set by configuration across bands of contiguous priorities
  - Protected objects can be used for tasks to communicate across bands
  - Tasks do not move across bands

A real-time component model

- Real-time components are objects
  - Instances of classes
  - Internal state + interfaces
  - Based on a reduced set of archetypes
    - Cyclic & sporadic tasks
    - Protected data
    - Passive data

To ensure consistent temporal behavior

- Two complementary approaches
- Static WCET analysis and response-time analysis can be used to ascertain correct temporal behavior at design time
- Platform mechanisms can be used at run time to ensure that temporal behavior stays within the boundaries asserted during analysis
  - Clocks, timers, timing events, …

Run-time services

- The run-time environment must provide services that help preserve properties asserted in the analysis
  - Real-time clocks & timers
  - Execution-time clocks & timers
  - Predictable scheduling
- We assume the execution environment to implement the Ravenscar tasking model
  - Ada 2005 with the Ravenscar profile
  - Augmented with (restricted) execution-time timers

Component structure



Taxonomy of components

- Cyclic component
- Sporadic component
- Protected (data) component
- Passive component

Cyclic component

- Release event from clock with fixed rate
- Real-time attributes
  - Period
  - Deadline
  - Worst-case execution time
- The most basic cyclic component does not need the control agent
  - The system clock delivers the activation event
  - The component behavior is fixed and immutable

Cyclic component (basic variant)



Cyclic component thread (spec)

```
task type Cyclic_Thread
   (Thread_Priority : Priority;
    Period          : Positive) is
  pragma Priority(Thread_Priority);
end Cyclic_Thread;
```

ms

cannot be Time_Span!

Cyclic component thread (body)

```
task body Cyclic_Thread is
  Next_Time : Time := <Start_Time>; --  taken at elaboration time
                                    --+ higher in the system
                                    --+ hierarchy
begin
  loop
    delay until Next_Time; -- so that all tasks start at T0
    OPCS.Cyclic_Operation; -- fixed and parameterless
    Next_Time := Next_Time + Milliseconds(Period);
  end loop;
end Cyclic_Thread;
```


Sporadic component – 1

- Release event from software-mediated event
  - Signaled by software or hardware interrupts
- Real-time attributes
  - Minimum inter-arrival time
  - Deadline
  - Worst-case execution time
- The control agent of the target component is used to signal the activation event
  - And to store-and-forward signal-related data (if any)

Sporadic component – 2



Sporadic component (spec)

```
task type Sporadic_Thread(Thread_Priority : Priority) is
  pragma Priority(Thread_Priority);
end Sporadic_Thread;

protected type OBCS(Ceiling : Priority) is
  pragma Priority(Ceiling);
  procedure Signal;
  entry     Wait;
private
   Occurred : Boolean := False;
end OBCS;
```

The sporadic thread is activated by calling the Signal operation

Sporadic component thread (body)

```
task body Sporadic_Thread is
   Next_Time : Time := <Start_Time>;
begin
   delay until Next_Time; -- so that all tasks start at T0
   loop
      OBCS.Wait;
      OPCS.Sporadic_Operation;
      --  may take parameters if they were delivered by Signal
      --+ and retrieved by Wait
   end loop;
end Sporadic_Thread;
```


Sporadic component control agent (body)

```
protected body OBCS is
   procedure Signal is
   begin
      Occurred := True;
   end Signal;
   entry Wait when Occurred is
   begin
      Occurred := False;
   end Wait;
end OBCS;
```


Other basic components

- **Protected component**
  - No thread, only synchronization and operations
  - Straightforward direct implementation with protected object
- **Passive component**
  - Purely functional behavior, neither thread nor synchronization
  - Straightforward direct implementation with functional package

Temporal properties

- Basic patterns only guarantee periodic or sporadic activation
- They must be augmented to guarantee additional temporal properties at run time
  - Minimum inter-arrival time for sporadic events
  - Deadline for all types of thread
  - WCET budgets for all types of thread

Minimum inter-arrival time

- Violations of the specified separation interval may increase interference on lower priority tasks and cause them to miss deadlines
- We must prevent sporadic thread from being activated earlier than stipulated
  - Compute earliest (absolute) allowable activation time
  - Withhold activation (if signaled) until that time

Sporadic component thread (spec)

```
task type Sporadic_Thread
   (Thread_Priority : Priority;
    Separation      : Positive) is
  pragma Priority(Thread_Priority);
end Sporadic_Thread;
```

ms

Minimum inter-arrival time expressed in ms

Sporadic component thread (body)

```
task body Sporadic_Thread is
  Release_Time : Time;
  Next_Release : Time := <Start_Time>;
begin
  loop
    delay until Next_Release;
    OBCS.Wait;
    Release_Time := Clock;
    OPCS.Sporadic_Operation;
    Next_Release := Release_Time + Milliseconds(Separation);
  end loop;
end Sporadic_Thread;
```

Still a single point of activation

Observation

- This code pattern may incur some temporal drift as the clock is read after task release
  - Hence preemption may hit just after the release but before reading the clock
  - The net effect is a larger separation than required
- It is better to read the clock at the place and time the task is released
  - Within the control agent
    - Which is protected and thus less exposed to general interference

Enhanced sporadic pattern – 1

```
protected type OBCS(Ceiling : Priority) is
  pragma Priority(Ceiling);
  procedure Signal;
  entry Wait(Release_Time : out Time);
private
   Occurred : Boolean := False;
end OBCS;
```


Enhanced sporadic pattern – 2

```
protected body OBCS is
  procedure Signal is
  begin
    Occurred := True;
  end Signal;

  entry Wait(Release_Time : out Time) when Occurred is
  begin
    Release_Time := Clock;
    Occurred := False;
  end Wait;
end OBCS;
```


Enhanced sporadic pattern – 3

```
task body Sporadic_Thread is
  Release_Time : Time;
  Next_Release : Time := <Start_Time>;
begin
  loop
    delay until Next_Release;
    OBCS.Wait(Release_Time);
    OPCS.Sporadic_Operation;
    Next_Release := Release_Time + Milliseconds(Separation);
  end loop;
end Sporadic_Thread;
```

Deadline overruns

- Deadline overruns in a task may occur as a result of
  - Higher priority tasks executing more often than expected
    - Prevented with inter-arrival time enforcement
  - Execution time of the same or higher priority tasks longer than stipulated
    - Programming errors
      - Bounding assertions violated by functional code
    - Inaccurate WCET analysis

Detection of deadline overruns

- Deadline overruns can be detected at run time with the help of timing events
  - A mechanism for requiring some application-level action to be executed at a given time
  - Timing events can only exist at library level under the Ravenscar profile
    - Statically allocated
- Minor enhancement possible for periodic tasks
  - Which however breaks the symmetry of code patterns

Enhanced cyclic pattern – 1

```
task type Cyclic_Thread
   (Thread_Priority : Priority;
    Period          : Positive;
    Deadline        : Positive) is
  pragma Priority(Thread_Priority);
end Cyclic_Thread;
```

Enhanced cyclic pattern – 2

```
Deadline_Overrun : Timing_Event; -- static, local per component
task body Cyclic_Thread is
  Next_Time : Time := <Start_Time>;
  Canceled  : Boolean := False;
begin
  loop
    delay until Next_Time;
    Set_Handler(Deadline_Overrun,
                Next_Time + Milliseconds(Deadline),
                Deadline_Overrun_Handler); -- application-specific
    OPCS.Cyclic_Operation;
    Cancel_Handler(Deadline_Overrun, Canceled);
    Next_Time := Next_Time + Milliseconds(Period);
  end loop;
end Cyclic_Thread;
```


Enhanced cyclic pattern – 3

```
Deadline_Overrun : Timing_Event; -- static, local per component
task body Cyclic_Thread is
  Next_Time : Time := <Start_Time>;
  Canceled  : Boolean := False;
begin
  loop
    -- setting again cancels any previous event
    Set_Handler(Deadline_Overrun,
                Next_Time + Milliseconds(Deadline),
                Deadline_Overrun_Handler); -- application-specific
    delay until Next_Time;
    OPCS.Cyclic_Operation;
    Next_Time := Next_Time + Milliseconds(Period);
  end loop;
end Cyclic_Thread;
```


Enhanced sporadic pattern – 4

```
task type Sporadic_Thread
   (Thread_Priority : Priority;
    Separation      : Positive;
    Deadline        : Positive) is
  pragma Priority(Thread_Priority);
end Sporadic_Thread;
```

Enhanced sporadic pattern – 5

```
Deadline_Overrun : Timing_Event; -- static, local per component
task body Sporadic_Thread is
  Release_Time    : Time;
  Next_Release    : Time := <Start_Time>;
  Canceled        : Boolean := False;
begin
  loop
    delay until Next_Release;
    OBCS.Wait(Release_Time);
    Set_Handler(Deadline_Overrun,
                Release_Time + Milliseconds(Deadline),
                Deadline_Overrun_Handler); -- application-specific
    OPCS.Sporadic_Operation;
    Cancel_Handler(Deadline_Overrun, Canceled);
    Next_Release := Release_Time + Milliseconds(Separation);
  end loop;
end Sporadic_Thread;
```

The deadline cannot be computed until returning from Wait

Execution-time overruns

- Tasks may execute for longer than stipulated, owing to programming errors
  - Bounding assertions violated by functional code
- WCET values used in feasibility analysis may be inaccurate
  - Optimistic instead of pessimistic
- WCET overruns can be detected at run time with the help of execution-time timers
  - Not included in Ravenscar
  - Extended profile

Enhanced cyclic pattern (spec) – 4

```
task type Cyclic_Thread
   (Thread_Priority : Priority;
    Period          : Positive;
    WCET_Budget     : Positive) is
  pragma Priority(Thread_Priority);
end Cyclic_Thread;
```

ms

Enhanced cyclic pattern (body) – 5

```
task body Cyclic_Thread is
  Next_Time : Time := <Start_Time>;
  Id : aliased constant Task_ID := Current_Task;
  WCET_Timer : Timer(Id'access);
begin
  loop
    delay until Next_Time;
    Set_Handler(WCET_Timer,
                Milliseconds(WCET_Budget),
                WCET_Overrun_Handler); -- application-specific
    OPCS.Cyclic_Operation;
    Next_Time := Next_Time + Milliseconds(Period);
  end loop;
end Cyclic_Thread;
```


Observations

- WCET overruns in sporadic tasks can be detected similarly
  - The timer should be set after the activation
  - There is no need for timer cancellation

Fault handling strategies

- Error logging
  - Only for low-criticality tasks
- Second chance
  - Use slack time and try to complete
- Mode change
  - Switch to *safe mode*
    - *Fail safe* or *fail soft* behaviour
  - How?

Fault handling scheme



Modifiers

- Cyclic and sporadic objects may have modifier operations
  - Mode change, behavior modifications, etc.
- ATC not allowed in Ravenscar
  - Modifier requests are queued in the OBCS
    - Control agent now required for cyclic components as well
  - The thread fetches requests from the OBCS queue and executes them whenever possible
  - "When" is determined by the adopted service policy

Cyclic thread with modifier

```
task body Cyclic_Thread is
  Next_Release_Time : Time := <Start_Time>;
  Request : Request_Type;
begin
  loop
    delay until Next_Release_Time;
    OBCS.Get_Request(Request); -- may include operation parameters
    case Request is
      when NO_REQ  => OPCS.Periodic_Activity;
      when ATC_REQ => -- may take parameters
                      OPCS.Modifier_Operation;
    end case;
    Next_Release_Time := Next_Release_Time + Period;
  end loop;
end Cyclic_Thread;
```


Control agent – 1

```
-- for cyclic thread
protected type OBCS (Ceiling: Priority) is
  pragma Priority(Ceiling);
  procedure Put_Request(Request : Request_Type);
  procedure Get_Request(out Request : Request_Type);
private
  Buffer : Request_Buffer; -- bounded queue
end OBCS;
```


Control agent – 2

```
-- for cyclic thread
protected body OBCS(Ceiling : Priority) is
  procedure Put_Request(Request : Request_Type) is
  begin
    Buffer.Put(Request);
  end Put_Request;

  procedure Get_Request(out Request : Request_Type) is
  begin
    if Buffer.Empty then
      Request := NO_REQ;
    else
      Buffer.Get(Request);
    end if;
  end Get_Request;
end OBCS;
```


Ada 2005 compilation chain

- Ada 2005 compiler and linker
  - Full support of Annex D – Real-time systems
- Real-time kernel
  - Implements the Ravenscar tasking model
- Ada run-time system
  - Implements the Ada tasking model on top of the kernel

GNAT for LEON

- Cross-compilation system targeted to LEON2 micro-processors
  - Radiation-hardened SPARC v8
  - ESA standard
- Components
  - GNAT Ada 2005 compiler (AdaCore)
  - GNARL run-time system (AdaCore)
  - ORK+ kernel (UPM)

GNAT compiler

- Ada 2005 cross-compilation system
  - Hosted on GNU/Linux
  - Targeted to ELF-SPARC v8
    - Real hardware or simulators
- Current version: GNAT GPL 2009
  - Supports Ada 2005
  - Ported to LEON2 at UPM
    - Including Ravenscar run-time system

ORK+

- Lightweight real-time kernel for the Ravenscar tasking model
- Evolution of ORK
  - Developed at UPM under ESA contract
- New Ada 2005 features
  - Timing events
  - Execution-time clocks and timers
  - Group budgets

ORK+ architecture



Compilation process



Cross-compilation and debugging



Running programs

- LEON2 simulator on host platform
  - E.g. TSIM
- LEON2 computer board
- Can use the GNAT programming system for cross-compilation, execution and debugging on target

Example

```
$ sparc-elf-gnatmake hello
[ $ sparc-elf-gnatmake -g hello -largs -Wl,-Map=hello.map ]
$ tsim –gdb
 ...
gdb interface: using port 1234
[ on another terminal – local or remote ]
$ sparc-elf-gdb hello
(gdb) target extended-remote 127.0.0.1:1234
 ...
(gdb) load
 ...
(gdb) cont
 ...
(gdb) detach
 ...
```


GNAT Programming System



Other tools

- Response time analysis
  - MAST (University of Cantabria, Spain)
  - http://mast.unican.es/
- Execution-time (WCET) analysis
  - Bound-T (Tidorum, Finland)
    - Static analysis
    - http://www.tidorum.fi/bound-t/
  - RapiTime (Rapita Systems, UK)
    - Measurement based
    - http://www.rapitasystems.com/rapitime

Summary

- A concrete tasking model
- Preservation of analysis assumptions and properties
- Execution time measurement
- Execution time timers
- Group budget (for servers)
- Timing events
- A real-time component model
- Tools